
[bookmark: foreword]

* * * First Change * * * *
[bookmark: _Toc153887922]5.5.2.1	N32-c connection establishment using HTTP CONNECT
5.5.2.1.A	General
This clause specifies the requirements that apply in scenarios where a PLMN SEPP makes use of Roaming Intermediaries and support messages generated by Roaming Intermediaries as specified in clause 5.9.3.2a of 3GPP TS 33.501 [6].
Editor's note: this clause will describe requirements regarding the establishment of the N32-c connection via one or two Roaming Intermediaries, using the HTTP CONNECT method.
Editor's note: Details about the HTTP CONNECT procedures are FFS. 
Editor's note: It is FFS how a Roaming Intermediary identifies the source PLMN in the HTTP CONNECT request.
The N32 Handshake procedures specified in clause 5.2 shall apply between the c-SEPP and p-SEPP with the following additions:
-	Prior to establishing the (end-to-end) N32-c TLS connection with the pSEPP, the c-SEPP shall send an HTTP CONNECT Request to the Roaming Intermediary to request the Roaming Intermediary to set up a TCP connection towards the p-SEPP as specified in clause 5.5.2.1.B (with one Roaming Intermediary between the SEPPs) and clause 5.5.2.1.C (with two Roaming Intermediaries between the SEPPs); upon receipt of a successful HTTP CONNECT response, the c-SEPP shall establish the N32-c TLS connection and proceed with the N32 Handshake procedures; 
-	If a Roaming Intermediary disallows the establishment of the N32-c connection, the Roaming Intermediary shall reject the HTTP CONNECT request as specified in clause 5.5.2.2.

* * * Next Change * * * *
5.5.2.1.B	Successful N32-c connection establishment via one Roaming Intermediary
Figure 5.5.2.1.B-1 depicts the successful establishment of the N32-c connection between c-SEPP and p-SEPP via one Roaming Intermediary (RI-A).


[bookmark: _Hlk156985893]Figure 5.5.2.1.B-1: Successful N32-c connection establishment via one Roaming Intermediary
1.	The c-SEPP shall establish a TCP connection with the RI-A.
2.	The c-SEPP shall send an HTTP CONNECT request to the RI-A to request the RI-A to establish a TCP connection towards the p-SEPP. The authority of the HTTP CONNECT request shall contain the p-SEPP's FQDN.
The c-SEPP shall include the following information in the 3gpp-Connect-Req-Info header in the HTTP CONNECT request:
-	the HTTP connect purpose set to "n32c" to indicate that the TCP connection requested to be established is to set up an N32-c connection between the c-SEPP and p-SEPP; 
-	the c-SEPP's PLMN ID or SNPN ID; and
-	the c-SEPP's FQDN.
The c-SEPP may also include the following information in the HTTP CONNECT request in the 3gpp-Connect-Req-Info header:
-	the intended N32 purpose(s) of the N32 connection.
The HTTP CONNECT request shall not contain any content (i.e. payload).
3.	The Roaming Intermediary shall determine whether to (dis)allow the N32-c connection establishment based on its roaming contractual agreements and the following parameters:
-	the source PLMN ID or SNPN-ID; and 
-	the target PLMN ID or SNPN ID.
The Roaming Intermediary may also consider the following information for the above determination:
-	the HTTP connect purpose, e.g. the Roaming Intermediary may accept an HTTP CONNECT request only for the purpose of establishing a N32-c connection; and/or
-	the intended N32 purposes, e.g. the Roaming Intermediary may accept to establish the N32-c connection between the two PLMNs only for specific purposes.
4.	If the RI-A allows the establishment of the N32-c connection, it shall establish the TCP connection towards the p-SEPP.
5.	On successful processing of the request and establishment of the TCP connection towards the p-SEPP, the RI-A shall respond to the c-SEPP with a "200 OK" status code and may include the following information in the 3gpp-Connect-Resp-Info header in the HTTP CONNECT response:
-	the allowed N32 purposes for the N32 connection, which may be a subset of the N32 purposes signalled in the request; and/or.
-	the p-SEPP FQDN, if the Roaming Intermediary has overwritten the p-SEPP FQDN received from the c-SEPP that the c-SEPP should use for sending its N32-c requests to the p-SEPP.
6	The c-SEPP shall establish the (end to end) N32-c TLS connection with the p-SEPP.
7.	The c-SEPP shall proceed with the N32-c Handshake procedures specified in clause 5.2.

* * * Next Change * * * *
5.5.2.1.C	Successful N32-c connection establishment via two Roaming Intermediaries 
Figure 5.5.2.1.C-1 depicts the successful establishment of the N32-c connection between c-SEPPs and p-SEPP via two RI’s (RI-A and RI-B).


Figure 5.5.2.1.C-1: Successful N32-c connection establishment via 2 Roaming Intermediaries
[bookmark: _Hlk156913583]Steps 1 to 3: same as steps 1 to 3 of Figure 5.5.2.1.B-1.4. If the RI-A allows the establishment of the N32-c connection, it shall establish a TCP connection towards the RI-B and send an HTTP CONNECT request to the RI-B, as described in for step 2 but with the following modification:
-	the RI-A shall include the RI-A's FQDN in the 3gpp-Connect-Req-Info header, instead of including the c-SEPP's FQDN.
steps 5 and 6: same as steps 3 and 4 of Figure 5.5.2.1.B-1.
Steps 7a and 7b: same as step 5 of Figure 5.5.2.1.B-1. The RI-A shall send the response to the c-SEPP only upon receiving the 200 OK response from the RI-B.
Steps 8 and 9: same as steps 6 and 7 of Figure 5.5.2.1.B-1.


* * * Next Change * * * *
X	Usage of HTTP CONNECT for N32-c connection establishment via Roaming Intermediaries
X.1	General
HTTP/2, as defined in IETF RFC 9113 [7], shall be used.
The Transmission Control Protocol as described in IETF RFC 793 [11] shall be used as transport protocol as required by HTTP/2 (see IETF RFC 9113 [7]).
The HTTP CONNECT method shall be used as defined in clause 9.3.6 of IETF RFC 9110 [9] and clause 8.5 of IETF RFC 9113 [7].
The ":authority" pseudo-header field in the HTTP/2 CONNECT request message shall be set to:
	":authority" = uri-host [":" port] as specified in clause 8.3.1 of IETF RFC 9113 [7].
The Uri-host shall indicate the FQDN of the p-SEPP.
The HTTP CONNECT request and the 200 OK status response shall not contain any content.
X.2	HTTP standards headers
The HTTP standard headers defined in Table X.2-1 shall be supported on the interface between the SEPP and Roaming Intermediary. 
Mandatory to support HTTP standard headers does not mean that all the HTTP requests and responses carry the identified request and response headers respectively. It only means it is mandatory to support the processing of the identified headers in request and response message.

[bookmark: _CRTable5_2_2_22]Table X.2-1: Mandatory to support HTTP response standard headers
	Name
	Reference
	Description

	Via
	IETF RFC 9110 [9]
	This header should be inserted by a Roaming Intermediary when relaying an HTTP error response (see clause 6.10.8 of 3GPP TS 29.500 [4]). It may be inserted when relaying other HTTP responses.
When inserted by a Roaming Intermediary, the received-protocol portion of the header field value should be set to "HTTP/2.0" or "2.0" and the received-by portion of the header field value should be formatted as follows:
- "RI-<RI FQDN>" for a Roaming Intermediary 

	Server
	IETF RFC 9110 [9]
	This header should be inserted by the originator of an HTTP error response (see clause 6.10.8 of 3GPP TS 29.500 [4]). It may be inserted otherwise.
When inserted by a Roaming Intermediary, the pattern of the header should be formatted as follows:
- "RI-<RI FQDN>" for a Roaming Intermediary

	NOTE:	The inclusion of the Via and Server header in an HTTP CONNECT error response message enables the receiving SEPP to determine which Roaming Intermediary has rejected the request in scenarios with two roaming intermediaries between the SEPPs.



X.3	HTTP custom headers
X.3.1	3gpp-Connect-Req-Info
The header enables to convey information in the HTTP CONNECT request to the Roaming Intermediary, that may be used by the Roaming Intermediary to determine whether to allow the establishment of the N32-c connection and/or for troiuble-shooting. 
The encoding of the header follows the ABNF as defined in IETF RFC 9110 [9].
[bookmark: _Hlk158030995]Connect-Req-Info-Header = "3gpp-Connect-Req-Info:" OWS "connect-purpose=" OWS connect-purpose-value";" OWS orig-network-id ";" OWS sender-fqdn [";" OWS intended-n32-purposes] *( ";" OWS req-param )

connect-purpose-value = "n32c" / token

orig-network-id = "originating-network-id=" OWS 3DIGIT "-" 2*3DIGIT [ "-" 11HEXDIG ]

sender-fqdn = "sender-fqdn=" OWS 4*( ALPHA / DIGIT / "-" / "." )

intended-n32-purposes = intended-n32-purpose *(";" OWS intended-n32-purpose)
intended-n32-purpose = "intended-n32-purpose=" OWS n32-purpose-value

n32-purpose-value = "ROAMING"
           / "INTER_PLMN_MOBILITY"
           / "SMS_INTERCONNECT"
           / "ROAMING_TEST"
           / "INTER_PLMN_MOBILITY_TEST"
           / "SMS_INTERCONNECT_TEST"
           / "SNPN_INTERCONNECT"
           / "SNPN_INTERCONNECT_TEST"
           / "DISASTER_ROAMING"
           / "DISASTER_ROAMING_TEST"
           / token

req-param = req-param-name "=" OWS req-param-value

req-param-name = token

req-param-value = token

EXAMPLE 1:	For an HTTP CONNECT request message from the c-SEPP to the RI-A to request establishing the TCP connection towards the p-SEPP:
3gpp-Connect-Req-Info: connect-purpose=n32c; originating-network-id=123-45;sender-fqdn=sepp12.5gc.mnc155.mcc400.3gppnetwork;intended-n32-purpose=ROAMING;intended-32-purpose=SMS_INTERCONNECT
EXAMPLE 2:	For an HTTP CONNECT request message from the RI-A to the RI-B to request establishing the TCP connection towards p-SEPP:
3gpp-Connect-Req-Info: connect-purpose=n32c;originating-network-id=123-45;sender-fqdn=ri234.rioperator.com;intended-n32-purpose=ROAMING;intended-n32-purpose=SMS_INTERCONNECT

X.3.2	3gpp-Connect-Resp-Info
The header enables to convey information in the HTTP CONNECT response towards the c-SEPP.
The encoding of the header follows the ABNF as defined in IETF RFC 9110 [9].
[bookmark: _Hlk158021273][bookmark: _Hlk157451412]Connect-Resp-Info-Header = "3gpp-Connect-Resp-Info:" OWS resp-param *( ";" OWS resp-param ) OWS

resp-param = allowed-n32-purposes / p-sepp-fqdn / other-resp-param

allowed-n32-purposes = allowed-n32-purpose *(OWS";"OWS allowed-n32-purpose)

allowed-n32-purpose = "allowed-n32-purpose=" n32-purpose-value

p-sepp-fqdn = "p-sepp-fqdn=" OWS 4*( ALPHA / DIGIT / "-" / "." )

other-resp-param = other-resp-param-name "=" OWS other-resp-param-value

other-resp-param-name = token

other-resp-param-value = token


EXAMPLE:	3gpp-Connect-Resp-Info: allowed-n32-purpose=ROAMING;allowed-n32-purpose=SMS_INTERCONNECT;p-sepp-fqdn=sepp2.5gc.mnc203.mcc422.3gppnetwork.org
* * * Next Change * * * *
[bookmark: _Toc153804012]Annex F (Normative):
ABNF grammar for HTTP custom headers for HTTP CONNECT
[bookmark: _Toc153804013]F.1	General
This Annex contains a self-contained set of ABNF rules, comprising the re-used rules from IETF RFCs, and the rules defined by the 3GPP custom headers defined in this specification (see clause X.3).
This grammar may be used as input to existing tools to help implementations to parse 3GPP custom headers.
F.2	ABNF definitions (Filename:"TS29573_CustomHeaders.abnf")
; ----------------------------------------
;   RFC 5234
; ----------------------------------------

HTAB   = %x09 ; horizontal tab

SP     = %x20

DIGIT  = %x30-39 ; 0-9

ALPHA  = %x41-5A / %x61-7A ; A-Z / a-z

HEXDIG = DIGIT / "A" / "B" / "C" / "D" / "E" / "F"

; ----------------------------------------
;   RFC 9110
; ----------------------------------------

OWS            = *( SP / HTAB )

RWS            = 1*( SP / HTAB )

tchar          = "!" / "#" / "$" / "%" / "&" / "'" / "*" / "+" / "-"
               / "." / "^" / "_" / "`" / "|" / "~" / DIGIT / ALPHA

token          = 1*tchar

; ----------------------------------------
;   3GPP TS 29.573
;
;   Version: 18.6.0 (March 2024)
;
;   (c) 2023, 3GPP Organizational Partners (ARIB, ATIS, CCSA, ETSI, TSDSI, TTA, TTC).
; ----------------------------------------

;
; Header: 3gpp-Connect-Req-Info
;
Connect-Req-Info-Header = "3gpp-Connect-Req-Info:" OWS "connect-purpose=" OWS connect-purpose-value ";" OWS orig-network-id ";" OWS sender-fqdn [";" OWS intended-n32-purposes] *( ";" OWS req-param )

connect-purpose-value =  "n32c" / token

orig-network-id = "originating-network-id=" OWS 3DIGIT "-" 2*3DIGIT [ "-" 11HEXDIG ]

sender-fqdn = "sender-fqdn=" OWS 4*( ALPHA / DIGIT / "-" / "." )

intended-n32-purposes = intended-n32-purpose *(";" OWS intended-n32-purpose)

intended-n32-purpose="intended-n32-purpose=" OWS n32-purpose-value

n32-purpose-value = "ROAMING"
           / "INTER_PLMN_MOBILITY"
           / "SMS_INTERCONNECT"
           / "ROAMING_TEST"
           / "INTER_PLMN_MOBILITY_TEST"
           / "SMS_INTERCONNECT_TEST"
           / "SNPN_INTERCONNECT"
           / "SNPN_INTERCONNECT_TEST"
           / "DISASTER_ROAMING"
           / "DISASTER_ROAMING_TEST"
           / token

req-param = req-param-name "=" OWS req-param-value

req-param-name = token

req-param-value = token

;
; Header: 3gpp-Connect-Resp-Info
;

Connect-Resp-Info-Header = "3gpp-Connect-Resp-Info:" OWS resp-param *( ";" OWS resp-param ) OWS

resp-param = allowed-n32-purposes / p-sepp-fqdn / other-resp-param

allowed-n32-purposes = allowed-n32-purpose *(OWS ";" OWS allowed-n32-purpose)

allowed-n32-purpose = "allowed-n32-purpose=" n32-purpose-value

p-sepp-fqdn = "p-sepp-fqdn=" OWS 4*( ALPHA / DIGIT / "-" / "." )

other-resp-param = other-resp-param-name "=" OWS other-resp-param-value

other-resp-param-name = token

other-resp-param-value = token

* * * End of Changes * * * *

